Integral of the product of two Gaussians

Luis Pedro Coelho

April 19, 2013

What is this?
Since there are a few hits of people who download these files directly from a Google results page, instead of from my website, I add here a small explanation:
These files contain derivations which I often use and, before I wrote them down cleanly, would often make small mistakes on (like leaving off a minus sign or such). They are also automatically checked with **sage**, so that the computer assures that there were no mis-steps in the derivation.
This is also why you see code in the derivation instead of just the equations. This is run by **sage**.

```
x = var('x')
mu1 = var('mu1')
mu2 = var('mu2')
s1 = var('s1', latex_name=r'\sigma_1^2', domain='positive')
s2 = var('s2', latex_name=r'\sigma_2^2', domain='positive')
assume(s1 > 0)
assume(s2 > 0)

def Z(s):
    return sqrt(2*pi*s)

def N(x, m, s):
    return 1./Z(s) * exp(- (x-m)^2 /(2*s))

product = N(x, mu1, s1) * N(x,mu2,s2)
```

We want to be able to compute:

$$\int_{-\infty}^{\infty} \frac{0.500000000000000\, e^{\left(-\frac{(\mu_2-x)^2}{2\,\sigma_2^2} - \frac{(\mu_1-x)^2}{2\,\sigma_1^2}\right)}}{\sqrt{\pi\sigma_1^2}\sqrt{\pi\sigma_2^2}}\,dx, \tag{1}$$

or, in sage:

```
Nint = integral(product,x,-infinity,infinity)
```

I assert that this is equal to:

```
s12 = s1*s2/(s1+s2)
Ndirect = Z(s12)/(Z(s1)*Z(s2)) * exp(- (mu1-mu2)^2/2/(s1+s2))
```

where

$$\sigma_{12}^2 = \frac{\sigma_1^2\sigma_2^2}{\sigma_1^2+\sigma_2^2} \tag{2}$$

$$N = \frac{\sqrt{\frac{\pi\sigma_1^2\sigma_2^2}{\sigma_1^2+\sigma_2^2}}\sqrt{2}e^{\left(-\frac{(\mu_1-\mu_2)^2}{2\,(\sigma_1^2+\sigma_2^2)}\right)}}{2\sqrt{\pi\sigma_1^2}\sqrt{\pi\sigma_2^2}} \tag{3}$$

Let us check the ratio again

```
ratio = (Nint/Ndirect)
ratio = ratio.simplify_full()
```

The ratio is 1.

We can also write the function above as

```
Ngaussian1 = Z(s12)*Z(s1+s2)/(Z(s1)*Z(s2)) *N(mu1,mu2,s1+s2)
```

The products of all the Zs is going to simplify to 1:

```
Zs = Z(s12)*Z(s1+s2)/(Z(s1)*Z(s2))
Zs = Zs.simplify_full()
```

Results in 1.

So, we get our final result:

```
Ngaussian = N(mu1,mu2,s1+s2)
ratio = (Nint/Ngaussian)
ratio = ratio.simplify_full()
```

The ratio is, again, 1.

Therefore:

$$\int N(x|\mu_1,\sigma_1^2)N(x|\mu_2,\sigma_2^2)dx = N(\mu_1|\mu_2,\sigma_1^2+\sigma_2^2) = N(\mu_2|\mu_1,\sigma_1^2+\sigma_2^2). \tag{4}$$